



Cooking with CQL: or How to Express eCQMs using CQL, QDM, and FHIR® Webinar

Supplemental Slides for the April 23, 2020 Session – FHIR and CQL

Shanna Hartman
Centers for Medicare & Medicaid Services
and
Bryn Rhodes, ESAC, Inc.













Agenda

- Fast Healthcare Interoperability Resources (FHIR®) and Clinical Quality Language (CQL)

FHIR and CQL

- Primitives
- Choice Types
- Slices
- Extensions

Patient Resource

Name	Flags	Card.	Type	Descrip
 Patient	N		DomainResource	Informal Element: modifier
 identifier	Σ	0..*	Identifier	An ident
 active	?! Σ	0..1	boolean	Whether
 name	Σ	0..*	HumanName	A name
 telecom	Σ	0..*	ContactPoint	A contac
 gender	Σ	0..1	code	male fe Administ
 birthDate	Σ	0..1	date	The date
 deceased[x]	?! Σ	0..1		Indicate:
 deceasedBoolean			boolean	
 deceasedDateTime			dateTime	
 address	Σ	0..*	Address	An addre
 maritalStatus		0..1	CodeableConcept	Marital (

- Elements
 - name
 - cardinality
 - type

Patient in CQL

```
20  define TestPrimitives:
21      Patient P
22          where P.gender.value = 'male'
23              and P.active.value is true
24              and P.birthDate.value before Today()
25              and ToConcept(P.maritalStatus) in "Marital Status"
26              and ToConcept(P.maritalStatus) ~ "Marital Status - Married"
27
28  define TestChoice:
29      Patient P
30          where (P.deceased as FHIR.boolean).value is false
31              or (P.deceased as FHIR.dateTime).value before Today()
```

BP

Constraints Slices


















<http://hl7.org/fhir/bp.html>

Name	Flags	Card.	Type	Description & Constraints
Observation		0..*		FHIR Blood Pressure Profile
code		1..1	CodeableConcept	Blood Pressure
coding			(Slice Definition)	Slice: Unordered, Open by value:code, value:system
coding:BPCode		1..1	Coding	
system		1..1	uri	Fixed Value: http://loinc.org
code		1..1	code	Fixed Value: 85354-9
valueQuantity		0..0		
component			(Slice Definition)	Slice: Unordered, Open by value:code.coding.code, va
component:SystolicBP		1..1	BackboneElement	
code		1..1	CodeableConcept	
coding			(Slice Definition)	Systolic Blood Pressure Slice: Unordered, Open by value:code, value:system
coding:SBPCode		1..1	Coding	Systolic Blood Pressure
system		1..1	uri	Fixed Value: http://loinc.org
code		1..1	code	Fixed Value: 8480-6
valueQuantity		0..1	Quantity	
value	S	1..1	decimal	
unit	S	1..1	string	
system	S	1..1	uri	Fixed Value: http://unitsofmeasure.org
code	S	1..1	code	Coded responses from the common UCUM units for vit Fixed Value: mm[Hg]
component:DiastolicBP		1..1	BackboneElement	
code		1..1	CodeableConcept	
coding			(Slice Definition)	Diastolic Blood Pressure Slice: Unordered, Open by value:code, value:system
coding:DBPCode		1..1	Coding	Diastolic Blood Pressure
system		1..1	uri	Fixed Value: http://loinc.org
code		1..1	code	Fixed Value: 8462-4
valueQuantity		0..1	Quantity	
value	S	1..1	decimal	
unit	S	1..1	string	
system	S	1..1	uri	Fixed Value: http://unitsofmeasure.org
code	S	1..1	code	Coded responses from the common UCUM units for vit Fixed Value: mm[Hg]

BP in FHIR

```
33 define TestSlices:  
34   [Observation: "Blood pressure"] BP  
35   let  
36     SystolicBP: singleton from (BP.component C where ToConcept(C.code) ~ "Systolic blood pressure"),  
37     DiastolicBP: singleton from (BP.component C where ToConcept(C.code) ~ "Diastolic blood pressure")  
38   where ToQuantity(SystolicBP.value as FHIR.Quantity) < 140 'mm[Hg]'  
39     and ToQuantity(DiastolicBP.value as FHIR.Quantity) < 90 'mm[Hg]'
```

Extensions (US Core Patient)

Name	Flags	Card.	Type	Description & Constraints
 Patient		0..*		
 us-core-race		0..1	(Complex)	US Core Race Extension URL: http://hl7.org/fhir/us/core/StructureDefinition/us-core-race
 us-core-ethnicity		0..1	(Complex)	US Core ethnicity Extension URL: http://hl7.org/fhir/us/core/StructureDefinition/us-core-ethnicity
 us-core-birthsex		0..1	code	Extension URL: http://hl7.org/fhir/us/core/StructureDefinition/us-core-birthsex Binding: Birth Sex (required)
 identifier		1..*	Identifier	
 system		1..1	uri	
 value		1..1	string	The value that is unique within the system.
 name	 I	1..*	HumanName	us-core-8: Patient.name.given or Patient.name.family or both SHALL be present
 family	 I	0..1	string	

Extensions in CQL

```
41 define TestSimpleExtensions:
42   Patient P
43     let birthsex: singleton from (P.extension E
44       where E.url.value = 'http://hl7.org/fhir/us/core/StructureDefinition/us-core-birthsex')
45     where (birthsex.value as FHIR.code).value = 'M'
46
47 define TestComplexExtensions:
48   Patient P
49     let
50       race: singleton from (P.extension E
51         where E.url.value = 'http://hl7.org/fhir/us/core/StructureDefinition/us-core-race'),
52       ombCategory: race.extension E where E.url.value = 'ombCategory',
53       detailed: race.extension E where E.url.value = 'detailed'
54   where (ombCategory 0 return ToConcept(O.value as FHIR.CodeableConcept)) contains "American Indian or Alaska Native"
55     and (detailed 0 return ToConcept(O.value as FHIR.CodeableConcept)) contains "Alaska Native"
```

FHIRHelpers

- A library of conversion functions for use with FHIR

```
3 using FHIR version '4.0.1'  
4  
5 include FHIRHelpers version '4.0.1'
```

System-defined Types

CQL Type	FHIR Type
Boolean	boolean
Integer	integer
Decimal	decimal
String	string
Date	date
DateTime	dateTime
Time	time
Quantity	Quantity
Ratio	Ratio
Code	Coding
Concept	CodeableConcept

Patient w/ Helpers

```
define TestPrimitives:  
  Patient P  
    where P.gender.value = 'male'  
      and P.active.value is true  
      and P.birthDate.value before Today()  
      and ToConcept(P.maritalStatus) in "Marital Status"  
      and ToConcept(P.maritalStatus) ~ "Marital Status - Married"
```

```
define TestPrimitives:  
  Patient P  
    where P.gender = 'male'  
      and P.active is true  
      and P.birthDate before Today()  
      and P.maritalStatus in "Marital Status"  
      and P.maritalStatus ~ "Marital Status - Married"
```

BP in FHIR with Helpers

```
33 define TestSlices:  
34   [Observation: "Blood pressure"] BP  
35   let  
36     SystolicBP: singleton from (BP.component C where ToConcept(C.code) ~ "Systolic blood pressure"),  
37     DiastolicBP: singleton from (BP.component C where ToConcept(C.code) ~ "Diastolic blood pressure")  
38   where ToQuantity(SystolicBP.value as FHIR.Quantity) < 140 'mm[Hg]'  
39     and ToQuantity(DiastolicBP.value as FHIR.Quantity) < 90 'mm[Hg]'
```

```
35 define TestSlices:  
36   [Observation: "Blood pressure"] BP  
37   let  
38     SystolicBP: singleton from (BP.component C where C.code ~ "Systolic blood pressure"),  
39     DiastolicBP: singleton from (BP.component C where C.code ~ "Diastolic blood pressure")  
40   where SystolicBP.value < 140 'mm[Hg]'  
41     and DiastolicBP.value < 90 'mm[Hg]'
```

Extensions with Helpers

```
define TestSimpleExtensions:
```

```
  Patient P
```

```
    let birthsex: singleton from (P.extension E
```

```
      where E.url.value = 'http://hl7.org/fhir/us/core/StructureDefinition/us-core-birthsex')
```

```
    where (birthsex.value as FHIR.code).value = 'M'
```

```
define TestSimpleExtensions:
```

```
  Patient P
```

```
    let birthsex: singleton from (P.extension E
```

```
      where E.url = 'http://hl7.org/fhir/us/core/StructureDefinition/us-core-birthsex')
```

```
    where birthsex.value = 'M'
```

Complex Extensions with Helpers

```
define TestComplexExtensions:
  Patient P
  let
    race: singleton from (P.extension E
      where E.url.value = 'http://hl7.org/fhir/us/core/StructureDefinition/us-core-race'),
    ombCategory: race.extension E where E.url.value = 'ombCategory',
    detailed: race.extension E where E.url.value = 'detailed'
  where (ombCategory 0 return ToConcept(O.value as FHIR.CodeableConcept)) contains "American Indian or Alaska Native"
    and (detailed 0 return ToConcept(O.value as FHIR.CodeableConcept)) contains "Alaska Native"
```

```
define TestComplexExtensions:
  Patient P
  let
    race: singleton from (P.extension E
      where E.url = 'http://hl7.org/fhir/us/core/StructureDefinition/us-core-race'),
    ombCategory: race.extension E where E.url = 'ombCategory',
    detailed: race.extension E where E.url = 'detailed'
  where (ombCategory 0 return O.value) contains "American Indian or Alaska Native"
    and (detailed 0 return O.value) contains "Alaska Native"
```

QICore

```
using QICore version '4.0.0'
```

```
define TestPrimitives:
```

```
  Patient P
```

```
    where P.gender = 'male'  
    and P.active is true  
    and P.birthDate before Today()  
    and P.maritalStatus in "Marital Status"  
    and P.maritalStatus ~ "Marital Status - Married"
```

```
define TestChoice:
```

```
  Patient P
```

```
    where P.deceased is false  
    or P.deceased before Today()
```


Slices in QICore

```
35 define TestSlices:  
36   [Observation: "Blood pressure"] BP  
37   let  
38     SystolicBP: singleton from (BP.component C where C.code ~ "Systolic blood pressure"),  
39     DiastolicBP: singleton from (BP.component C where C.code ~ "Diastolic blood pressure")  
40   where SystolicBP.value < 140 'mm[Hg]'  
41     and DiastolicBP.value < 90 'mm[Hg]'
```

```
29 define TestSlices:  
30   ["observation-bp"] BP  
31   where BP.SystolicBP.value < 140 'mm[Hg]'  
32     and BP.DiastolicBP.value < 90 'mm[Hg]'
```

Extensions in QICore

```
43  define TestSimpleExtensions:  
44    Patient P  
45      let birthsex: singleton from (P.extension E  
46        where E.url = 'http://hl7.org/fhir/us/core/StructureDefinition/us-core-birthsex')  
47        where birthsex.value = 'M'
```

```
34  define TestSimpleExtensions:  
35    Patient P  
36      where P.birthsex = 'M'
```

Complex Extensions in QICore

```
49 define TestComplexExtensions:
50   Patient P
51     let
52       race: singleton from (P.extension E
53         where E.url = 'http://hl7.org/fhir/us/core/StructureDefinition/us-core-race'),
54       ombCategory: race.extension E where E.url = 'ombCategory',
55       detailed: race.extension E where E.url = 'detailed'
56     where (ombCategory 0 return 0.value) contains "American Indian or Alaska Native"
57           and (detailed 0 return 0.value) contains "Alaska Native"
```

```
38 define TestComplexExtensions:
39   Patient P
40     where P.race.ombCategory contains "American Indian or Alaska Native"
41     and P.race.detailed contains "Alaska Native"
```



Resources

- HL7 Standard: Clinical Quality Language Specification, Release 1 STU2
 - <http://cql.hl7.org>
- HL7 CDS Workgroup Project Homepage
 - http://wiki.hl7.org/index.php?title=Clinical_Quality_Language
- GitHub Tools Repository
 - https://github.com/cqframework/clinical_quality_language
- eCQI Resource Center - <https://ecqi.healthit.gov>
 - CQL Space - <https://ecqi.healthit.gov/cql>



FHIR Resources

- HL7 Fast Healthcare Interoperability Resources (FHIR)
 - <http://hl7.org/fhir/>
- Quality Improvement Core (QI-Core) Implementation Guide
 - <http://hl7.org/fhir/us/qicore/>
- FHIR Quality Measure Implementation Guide
 - <http://hl7.org/fhir/us/cqfmeasures/>
- Example FHIR Measures (based on 2020 program year)
 - <http://hl7.org/fhir/us/cqfmeasures/>

Soliciting Examples for the Next Session

- Feedback Survey pop up at the close of this session
- ESAC CQL Email Account
 - cql-esac@esacinc.com
- ONC JIRA CQL Issue Tracker
 - <https://oncprojecttracking.healthit.gov/support/browse/CQLIT>